HEWLETT-PACKARD

HP-33E

MATHEMATICS

Applications



For Continuous Memory Models

Although this book refers specifically to the HP-33E or HP-38E, the programs and calculations contained herein apply equally well to the HP-33C or HP-38C. The user should note, however, that the display format and data register contents are retained by the calculator even though it has been turned off. It may be desirable to reset or clear these conditions before running programs or making calculations.

HEWLETT hp PACKARD

5955-5259

HEWLETT hp PACKARD

HP-33E

Mathematics Applications

February, 1978

00033-90030

Printed in U.S.A.

Introduction

This Mathematics Applications book was written to help you get the most from your HP-33E calculator. The programs were chosen to provide useful calculations for many of the common problems encountered in mathematics.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Mathematics Applications book will be a valuable tool in your work and would appreciate your comments about it.

Contents

Algebra and Number Theory

Quadratic Equation

The roots x_1, x_2 of

$$ax^2 + bx + c = 0$$

are given by

$$x_{1,2} = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}$$

If

$$D = (b^2 - 4ac)/4a^2$$

is positive or zero, the roots are real. In these cases, better accuracy may sometimes by obtained by first calculating the root with the larger absolute value:

If

$$-\frac{b}{2a} \geqslant 0, \qquad x_1 = -\frac{b}{2a} + \sqrt{D}$$

If

$$-\frac{b}{2a} < 0, \qquad x_1 = -\frac{b}{2a} - \sqrt{D}$$

In either case,

$$x_2 = \frac{c}{x_1\, a}$$

If $D < 0$, the roots are complex, being

$$u \pm iv = \frac{-b}{2a} \pm \frac{\sqrt{4ac - b^2}}{2a}\, i$$

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	x⇄y	19– 21
ENTER↑	01– 31	g x<0	20– 15 41
R↓	02– 22	GTO 24	21– 13 24
÷	03– 71	+	22– 51
2	04– 2	GTO 26	23– 13 26
÷	05– 71	x⇄y	24– 21
CHS	06– 32	−	25– 41
ENTER↑	07– 31	R/S	26– 74
g x^2	08– 15 0	g 1/x	27– 15 3
R↓	09– 22	RCL 0	28– 24 0
R↓	10– 22	×	29– 61
x⇄y	11– 21	GTO 00	30– 13 00
÷	12– 71	CHS	31– 32
STO 0	13– 23 0	f $\sqrt{x}$	32– 14 0
−	14– 41	x⇄y	33– 21
f PAUSE	15– 14 74	R/S	34– 74
g x<0	16– 15 41	x⇄y	35– 21
GTO 31	17– 13 31	GTO 00	36– 13 00
f $\sqrt{x}$	18– 14 0		

REGISTERS			
R_0 c/a	R_1	R_2	R_3
R_4	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Initialize		[f] [PRGM]	
3	Enter coefficients and			
	display D	c	[ENTER↑]	
		b	[ENTER↑]	
		a	[R/S]	(D)
4	If D ⩾ 0, roots are real			x_1
			[R/S]	x_2
	or			
	If D < 0, roots are complex			
	of form u ± iv			u
			[R/S]	v
5	For new case, go to step 3.			

Example:

1. $x^2 + x - 6 = 0$ ($D = 6.2500$, $x_1 = -3.0000$, $x_2 = 2.0000$)
2. $3x^2 + 2x - 1 = 0$ ($D = 0.4444$, $x_1 = -1.0000$, $x_2 = 0.3333$)
3. $2x^2 - 3x + 5 = 0$ ($D = -1.9375$, $x_1, x_2 = 0.7500 \pm 1.3919i$)

Keystrokes	Display
[f] [PRGM]	
6 [CHS] [ENTER↑] 1	
[ENTER↑] 1 [R/S]	-3.0000
[R/S]	2.0000
1 [CHS] [ENTER↑] 2	
[ENTER↑] 3 [R/S]	-1.0000
[R/S]	0.3333
5 [ENTER↑] 3 [CHS]	
[ENTER↑] 2 [R/S]	0.7500
[R/S]	1.3919

Complex Arithmetic (+, −, ×, ÷)

Let $a_1 + ib_1$ and $a_2 + ib_2$ be two complex numbers. The arithmetic operations +, −, ×, ÷ are defined as follows:

1. +, addition

$$(a_1 + ib_1) + (a_2 + ib_2) = (a_1 + a_2) + (b_1 + b_2)i$$

2. −, subtraction

$$(a_1 + ib_1) - (a_2 + ib_2) = (a_1 - a_2) + (b_1 - b_2)i$$

3. ×, multiplication

$$(a_1 + ib_1) \times (a_2 + ib_2) = r_1 \, r_2 \, e^{i(\theta_1 + \theta_2)}$$

4. ÷, division

$$\frac{(a_1 + ib_1)}{(a_2 + ib_2)} = \frac{r_1}{r_2} \, e^{i(\theta_1 - \theta_2)}, \; a_2 + ib_2 \neq 0$$

where $r_1 \, e^{i\theta_1}$ is the polar representation of $a_1 + ib_1$ and $r_2 \, e^{i\theta_2}$ is the polar representation of $a_2 + ib_2$. In each case let the answer be $x + iy$.

After a calculation is finished x is stored in R_0 as well as the X-register and y is stored in R_1 as well as the Y-register. In this way arithmetic operations can be chained together.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	g →P	17– 15 4
CHS	01– 32	STO 2	18– 23 2
x⇄y	02– 21	R↓	19– 22
CHS	03– 32	RCL 1	20– 24 1
x⇄y	04– 21	RCL 0	21– 24 0
RCL 0	05– 24 0	g →P	22– 15 4
+	06– 51	STO × 2	23– 23 61 2
x⇄y	07– 21	R↓	24– 22
RCL 1	08– 24 1	+	25– 51
+	09– 51	RCL 2	26– 24 2
GTO 29	10– 13 29	f →R	27– 14 4
g →P	11– 15 4	x⇄y	28– 21
g 1/x	12– 15 3	STO 1	29– 23 1
x⇄y	13– 21	x⇄y	30– 21
CHS	14– 32	STO 0	31– 23 0
x⇄y	15– 21	GTO 00	32– 13 00
GTO 18	16– 13 18		

REGISTERS			
R_0 a_1, x	R_1 b_1, y	R_2 Used	R_3
R_4	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Store first complex number	b_1	[STO] 1	
		a_1	[STO] 0	
3	Key in next number	b_2	[ENTER↑]	
		a_2		
4	For addition		[GSB] 05	$a_1 + a_2$
	or			
	subtraction		[GSB] 01	$a_1 - a_2$
	or			
	multiplication		[GSB] 17	$a_1 a_2 - b_1 b_2$
	or			
	division		[GSB] 11	Real Part
5	For imaginary part		[x⇄y]	Imaginary
6	For next calculation in chain,			
	go to step 3.			
7	For new case, go to step 2.			

Examples:

1. $(1.2 + 3.7i) - (2.6 - 1.9i) = -1.4 + 5.6i$

2. $\dfrac{3 + 4i}{7 - 2i} = 0.2453 + 0.6415i$

3. $\left[\dfrac{(3 + 4i) + (7.4 - 5.6i)}{(7 - 2i)}\right] [3.1 + 4.6i] = 3.6121 + 7.1577i$

Keystrokes	Display
3.7 [STO] 1	
1.2 [STO] 0	
1.9 [CHS] [ENTER↑] 2.6	
[GSB] 01	−1.4000
[x⇄y]	5.6000

Keystrokes	Display
4 [STO] 1	
3 [STO] 0	
2 [CHS] [ENTER↑] 7	
[GSB] 11	0.2453
[x⇄y]	0.6415
4 [STO] 1	
3 [STO] 0	
5.6 [CHS] [ENTER↑] 7.4	
[GSB] 05	
2 [CHS] [ENTER↑] 7	
[GSB] 11	
4.6 [ENTER↑]	
3.1 [GSB] 17	3.6121
[x⇄y]	7.1577

Complex Functions ($|z|$, z^n, and $z^{1/n}$)

A complex number $z = x + iy$ has polar representation $r\ e^{i\theta}$. The formulas used to evaluate the given functions are as follows:

$$|z| = r = \sqrt{x^2 + y^2}$$

$$z^n = r^n e^{in\theta} \qquad n = \pm(1, 2, 3, \ldots)$$

$$z^{1/n} = r^{1/n} e^{i\left(\frac{\theta}{n} + \frac{360k}{n}\right)}, k = 0, 1, \ldots, n-1$$

The answer is represented by $u + iv$

KEY ENTRY	DISPLAY
f CLEAR PRGM	00
RCL 2	01– 24 2
RCL 1	02– 24 1
g →P	03– 15 4
g RTN	04– 15 12
STO 3	05– 23 3
GSB 01	06– 12 01
RCL 3	07– 24 3
f y^x	08– 14 3
STO 5	09– 23 5
x⇄y	10– 21
RCL 3	11– 24 3
×	12– 61
STO 4	13– 23 4
GTO 38	14– 13 38
STO 3	15– 23 3
GSB 01	16– 12 01
RCL 3	17– 24 3
g 1/x	18– 15 3
f y^x	19– 14 3
x⇄y	20– 21

KEY ENTRY	DISPLAY
RCL 3	21– 24 3
÷	22– 71
3	23– 3
6	24– 6
0	25– 0
RCL 0	26– 24 0
×	27– 61
RCL 3	28– 24 3
÷	29– 71
+	30– 51
x⇄y	31– 21
f →R	32– 14 4
R/S	33– 74
1	34– 1
STO + 0	35– 23 51 0
RCL 3	36– 24 3
GTO 16	37– 13 16
RCL 5	38– 24 5
f →R	39– 14 4
GTO 00	40– 13 00

REGISTERS			
R_0 index	R_1 x	R_2 y	R_3 n
R_4 n θ	R_5 r^n	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Input the complex number			
	$z = x + iy$	x	[STO] 1	
		y	[STO] 2	
3	Select one of the 3 functions:			
	• Magnitude $\|z\|$		[GSB] 01	$\|z\|$
	• z^n	n	[GSB] 05	u
			[x⇄y]	v
	• $z^{1/n}$	0	[STO] 0	
		n	[GSB] 15	u_1
			[x⇄y]	v_1
			[R/S]	u_2
			[x⇄y]	v_2
			:	:
			:	:
			[R/S]	u_n
			[x⇄y]	v_n
4	For a new complex number			
	go to step 2.			

Examples:

1. $|12 - 5i| = 13.00$
2. $(6 - i)^2 = 35.00 - 12.00i$
3. $\frac{1}{2 + 5i} = 0.0690 - 0.1724i$
4. $\sqrt{3 + 4i} = \pm (2.00 + 1.00i)$

Keystrokes	Display
12 [STO] 1	
5 [CHS] [STO] 2	
[GSB] 01	13.0000
6 [STO] 1	
1 [CHS] [STO] 2	
2 [GSB] 05	35.0000
[x⇄y]	–12.0000
2 [STO] 1	
5 [STO] 2	
1 [CHS] [GSB] 05	0.0690
[x⇄y]	–0.1724
3 [STO] 1	
4 [STO] 2	
0 [STO] 0	
2 [GSB] 15	2.0000
[x⇄y]	1.0000
[R/S]	–2.0000
[x⇄y]	–1.0000

Determinant and Inverse of a 2 × 2 Matrix

Let $A = \begin{bmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{bmatrix}$ be a 2 × 2 matrix.

The determinant of A denoted by Det A or $|A|$ is evaluated by the following formula:

$$\text{Det } A = a_{22}\, a_{11} - a_{12}\, a_{21}$$

Also, the program evaluates the multiplicative inverse A^{-1} of A. The following formula is used: (A^{-1} exists only when $|A| \neq 0$).

$$A^{-1} = \begin{bmatrix} a_{22}/\text{Det } A & -a_{12}/\text{Det } A \\ -a_{21}/\text{Det } A & a_{11}/\text{Det } A \end{bmatrix} = A'$$

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
[f] CLEAR [PRGM]	00	[RCL] 2	14– 24 2
[RCL] 4	01– 24 4	[RCL] 0	15– 24 0
[RCL] 1	02– 24 1	[÷]	16– 71
[×]	03– 61	[CHS]	17– 32
[RCL] 2	04– 24 2	[R/S]	18– 74
[RCL] 3	05– 24 3	[RCL] 3	19– 24 3
[×]	06– 61	[RCL] 0	20– 24 0
[−]	07– 41	[÷]	21– 71
[STO] 0	08– 23 0	[CHS]	22– 32
[R/S]	09– 74	[R/S]	23– 74
[RCL] 4	10– 24 4	[RCL] 1	24– 24 1
[RCL] 0	11– 24 0	[RCL] 0	25– 24 0
[÷]	12– 71	[÷]	26– 71
[R/S]	13– 74	[GTO] 00	27– 13 00

REGISTERS			
R_0 Det A	R_1 a_{11}	R_2 a_{12}	R_3 a_{21}
R_4 a_{22}	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store matrix	a_{11}	[STO] 1	
		a_{12}	[STO] 2	
		a_{21}	[STO] 3	
		a_{22}	[STO] 4	
3	Calculate the determinant		[GSB] 01	Det A
4	Calculate the inverse		[R/S]	a_{11}'
			[R/S]	a_{12}'
			[R/S]	a_{21}'
			[R/S]	a_{22}'
5	For new case, go to step 2.			

Example:

Find the determinant and inverse of the matrix.

$$A = \begin{bmatrix} 3 & 2 \\ 4 & -4 \end{bmatrix}$$

Solution:

$$\text{Det } A = -20$$

$$A^{-1} = \begin{bmatrix} 0.20 & 0.10 \\ 0.20 & -0.15 \end{bmatrix}$$

Keystrokes	Display
3 [STO] 1	
2 [STO] 2	
4 [STO] 3	
4 [CHS] [STO] 4	
[GSB] 01	–20.0000
[R/S]	0.2000
[R/S]	0.1000
[R/S]	0.2000
[R/S]	–0.1500

Simultaneous Equations in Two Unknowns

Let $ax + by = e$

and $cx + dy = f$

be a system of two equations in two unknowns. Cramer's Rule is used to find the solution.

$$x = \frac{\begin{vmatrix} e & b \\ f & d \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}} = \frac{ed - bf}{ad - bc} \qquad y = \frac{\begin{vmatrix} a & e \\ c & f \end{vmatrix}}{\begin{vmatrix} a & b \\ c & d \end{vmatrix}} = \frac{af - ec}{ad - bc}$$

If $ad - bc = 0$ the calculator displays *Error 0*. In this case no solution or no unique solution exists.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	−	14– 41
RCL 3	01– 24 3	STO 0	15– 23 0
RCL 5	02– 24 5	÷	16– 71
×	03– 61	R/S	17– 74
RCL 2	04– 24 2	RCL 1	18– 24 1
RCL 6	05– 24 6	RCL 6	19– 24 6
×	06– 61	×	20– 61
−	07– 41	RCL 3	21– 24 3
RCL 1	08– 24 1	RCL 4	22– 24 4
RCL 5	09– 24 5	×	23– 61
×	10– 61	−	24– 41
RCL 2	11– 24 2	RCL 0	25– 24 0
RCL 4	12– 24 4	÷	26– 71
×	13– 61	GTO 00	27– 13 00

REGISTERS			
R_0 ad − bc	R_1 a	R_2 b	R_3 e
R_4 c	R_5 d	R_6 f	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store constants	a	[STO] 1	
		b	[STO] 2	
		e	[STO] 3	
		c	[STO] 4	
		d	[STO] 5	
		f	[STO] 6	
3	Find x and y		[GSB] 01	x
			[R/S]	y
4	For new case, go to step 2.			

Example:

$5x - 3y = 12$

$2x + y = 9$

Solution:

$x = 3.5455$

$y = 1.9091$

Keystrokes	Display
5 [STO] 1	
3 [CHS] [STO] 2	
12 [STO] 3	
2 [STO] 4	
1 [STO] 5	
9 [STO] 6 [GSB] 01	3.5455
[R/S]	1.9091

Number in Base b to Number in Base 10

This program consists of two subprograms. The first changes the integer part of a number in base b to a number in base 10.

$$I_{10} = i_n i_{n-1} \ldots i_2 i_1 = i_n b^{n-1} + i_{n-1} b^{n-2} + \ldots + i_2 b + i_1$$

This is evaluated in the form

$$b(\ldots(b(b(i_n b + i_{n-1}) + i_{n-2}) + \ldots) + i_2) + i_1$$

The second subprogram changes the fraction part of a number in base b to a number in base 10.

$$F_{10} = f_1 f_2 \ldots f_m = f_1 b^{-1} + f_2 b^{-2} + \ldots + f_m b^{-m}$$

Together the two programs can convert any number in base b to a number in base 10. Zeros must be entered in their proper place.

KEY ENTRY	DISPLAY
f CLEAR PRGM	00
STO 1	01– 23 1
RCL 0	02– 24 0
ENTER↑	03– 31
ENTER↑	04– 31
ENTER↑	05– 31
RCL 1	06– 24 1
R/S	07– 74
STO 1	08– 23 1
CLX	09– 34
+	10– 51
×	11– 61
RCL 1	12– 24 1
+	13– 51

KEY ENTRY	DISPLAY
GTO 07	14– 13 07
RCL 0	15– 24 0
g 1/x	16– 15 3
STO 2	17– 23 2
STO 3	18– 23 3
×	19– 61
R/S	20– 74
RCL 2	21– 24 2
RCL 3	22– 24 3
×	23– 61
STO 3	24– 23 3
×	25– 61
+	26– 51
GTO 20	27– 13 20

REGISTERS			
R_0 b	R_1 Used	R_2 b^{-1}	R_3 b^{-j}
R_4	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store base	b	[STO] 0	
3	For integer part, input left			
	most digit	i_n	[GSB] 01	
4	Perform for j = n−1 .,,,. 2:			
	Input next digit	i_j*	[R/S]	
5	Input final digit	i_1*	[R/S]	I_{10}
6	For fractional part, input digit			
	after decimal	f_1	[GSB] 15	
7	Perform for i = 2 .,,,. m−1:			
	Input next digit	f_j*	[R/S]	
8	Input final digit	f_m*	[R/S]	F_{10}
9	For new case, go to step 2.			
	* The stack must be main-			
	tained at these points.			

Examples:

1. $1777_8 = 1023_{10}$
2. $143.2044_5 = 48.4384_{10}$

Keystrokes	Display
8 [STO] 0	
1 [GSB] 01	
7 [R/S] 7 [R/S]	
7 [R/S]	1,023.0000
5 [STO] 0	
1 [GSB] 01	
4 [R/S] 3 [R/S]	48.0000
2 [GSB] 15	
0 [R/S] 4 [R/S]	
4 [R/S]	0.4384

Number in Base 10 to Number in Base b

This program will convert any positive number in base 10, N_{10}, to a number in base b, N_b, where $2 \leq b \leq 100$. The algorithm used is an iterative one which adds one more digit to N_b at each iteration. The program pauses as each new N_b is calculated to display successive approximations to the final answer. When the displayed value of N_b has reached the accuracy desired by the user, he should press R/S to halt the program, then RCL 3 to display N_b.

Notes:

1. When the base b is such that $11 \leq b \leq 100$, two display positions are allocated to each digit of N_b. Begin partitioning to the right and to the left of the decimal point. For example, 41106.12 in base 16 stands for 4B6.C.

2. If the calculation is terminated with all 9's on the display, it means the machine is overflowed and the range of machine has been exceeded.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	GTO 23	20– 13 23
RCL 0	01– 24 0	g INT	21– 15 32
1	02– 1	GTO 26	22– 13 26
0	03– 0	g INT	23– 15 32
f x>y	04– 14 51	1	24– 1
GTO 11	05– 13 11	−	25– 41
f x=y	06– 14 71	STO 4	26– 23 4
GTO 11	07– 13 11	RCL 2	27– 24 2
1	08– 1	x⇄y	28– 21
0	09– 0	f y^x	29– 14 3
0	10– 0	RCL 3	30– 24 3
STO 2	11– 23 2	+	31– 51
0	12– 0	STO 3	32– 23 3
STO 3	13– 23 3	f PAUSE	33– 14 74
RCL 1	14– 24 1	f PAUSE	34– 14 74
f LN	15– 14 1	RCL 0	35– 24 0
RCL 0	16– 24 0	RCL 4	36– 24 4
f LN	17– 14 1	f y^x	37– 14 3
÷	18– 71	STO − 1	38– 23 41 1
g x<0	19– 15 41	GTO 14	39– 13 14

REGISTERS			
R_0 b	R_1 N_{10}	R_2 10 or 100	R_3 N_b
R_4 1 digit	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Set display format		[f] [FIX] 9	
3	Store base and decimal			
	number	b	[STO] 0	
		N_{10}	[STO] 1 [GSB] 01	
4	Display successive approxi-			
	mations to N_b.			(N_b)
5	When number is shown with			
	desired accuracy, press			
	[R/S] to halt, then		[RCL] 3	N_b
6	For new case, go to step 3.			

Examples:

1. $67.32_{10} = 403.050114_{16}$
   $= 43.51E_{16}$

2. $\pi = 3.141592654_{10} = 11.00100100_2$

Keystrokes	Display	
[f] [FIX] 9		
16 [STO] 0		
67.32 [STO] 1		
[GSB] 01	403.0501140	(Pause)
[R/S] [RCL] 3	403.0501140	
2 [STO] 0 [g] [π]		
[STO] 1		
[GSB] 01	11.00100100	(Pause)
[R/S] [RCL] 3	11.00100100	

Vector Cross Product

If $A = (a_1, a_2, a_3)$ and $B = (b_1, b_2, b_3)$ are two three dimensional vectors then the cross product of A and B is denoted by $A \times B$ and is calculated as follows:

$$A \times B = \left(\begin{vmatrix} a_2 & a_3 \\ b_2 & b_3 \end{vmatrix}, - \begin{vmatrix} a_1 & a_3 \\ b_1 & b_3 \end{vmatrix}, \begin{vmatrix} a_1 & a_2 \\ b_1 & b_2 \end{vmatrix} \right) =$$

$$(a_2 b_3 - a_3 b_2, a_3 b_1 - a_1 b_3, a_1 b_2 - a_2 b_1)$$

Let the solution be represented by (c_1, c_2, c_3).

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	RCL 6	13– 24 6
RCL 2	01– 24 2	×	14– 61
RCL 6	02– 24 6	−	15– 41
×	03– 61	R/S	16– 74
RCL 3	04– 24 3	RCL 1	17– 24 1
RCL 5	05– 24 5	RCL 5	18– 24 5
×	06– 61	×	19– 61
−	07– 41	RCL 2	20– 24 2
R/S	08– 74	RCL 4	21– 24 4
RCL 3	09– 24 3	×	22– 61
RCL 4	10– 24 4	−	23– 41
×	11– 61	GTO 00	24– 13 00
RCL 1	12– 24 1		

REGISTERS			
R_0	R_1 a_1	R_2 a_2	R_3 a_3
R_4 b_1	R_5 b_2	R_6 b_3	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Store A	a_1	[STO] 1	
		a_2	[STO] 2	
		a_3	[STO] 3	
3	Store B	b_1	[STO] 4	
		b_2	[STO] 5	
		b_3	[STO] 6	
4	Calculate cross product		[GSB] 01	c_1
			[R/S]	c_2
			[R/S]	c_3
5	For new case, go to step 2.			

Example:

Let A = (2, 5, 2)
B = (3, 3, −4).

Solution:

A × B = (−26, 14, −9)

Keystrokes	Display
[f] [FIX] 2	
2 [STO] 1	
5 [STO] 2	
2 [STO] 3	
3 [STO] 4	
3 [STO] 5	
4 [CHS] [STO] 6	
[GSB] 01	−26.00
[R/S]	14.00
[R/S]	−9.00

Angle Between, Norm, and Dot Product of Vectors

Let $\vec{a} = (a_1, a_2 \ldots, a_n)$ and $\vec{b} = (b_1, b_2, \ldots, b_n)$ be two vectors.

The norm of $\vec{a}$ is denoted by $|\vec{a}|$ and is calculated by the following formula:

$$|\vec{a}| = \sqrt{a_1{}^2 + a_2{}^2 + \ldots + a_n{}^2}$$

similarly,

$$|\vec{b}| = \sqrt{b_1{}^2 + b_2{}^2 + \ldots + b_n{}^2}$$

The dot product of $\vec{a}$ and $\vec{b}$ is denoted by $\vec{a} \cdot \vec{b}$ and is calculated by the following formula:

$$\vec{a} \cdot \vec{b} = a_1 b_1 + a_2 b_2 + \ldots + a_n b_n$$

The angle between $\vec{a}$ and $\vec{b}$ is denoted by θ and is calculated by the following formula:

$$\theta = \cos^{-1}\left(\frac{\vec{a} \cdot \vec{b}}{|\vec{a}| \cdot |\vec{b}|}\right)$$

The angle is calculated in any angular mode. When calculated in degrees, decimal degrees are assumed.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	STO + 2	11– 23 51 2
ENTER↑	01– 31	GTO 00	12– 13 00
g x^2	02– 15 0	RCL 2	13– 24 2
STO + 1	03– 23 51 1	RCL 0	14– 24 0
R↓	04– 22	RCL 1	15– 24 1
x⇄y	05– 21	×	16– 61
ENTER↑	06– 31	f $\sqrt{x}$	17– 14 0
g x^2	07– 15 0	÷	18– 71
STO + 0	08– 23 51 0	g $\cos^{-1}$	19– 15 8
R↓	09– 22	GTO 00	20– 13 00
×	10– 61		

REGISTERS			
R_0 Σa_i^2	R_1 Σb_i^2	R_2 $\Sigma a_i b_i$	R_3
R_4	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program.			
2	Initialize		f REG f PRGM	
3	Perform for i = 1, … n:			
	Key in a_i and b_i	a_i	ENTER↑	
		b_i	R/S	
4	Find norm of $\vec{a}$		RCL 0 f $\sqrt{x}$	$\lvert\vec{a}\rvert$
5	Find norm of $\vec{b}$		RCL 1 f $\sqrt{x}$	$\lvert\vec{b}\rvert$
6	Find $\lvert\vec{a} \cdot \vec{b}\rvert$		RCL 2	$\lvert\vec{a} \cdot \vec{b}\rvert$
7	Calculate angle between			
	$\vec{a}$ and $\vec{b}$		GSB 13	θ

Example:

Let $\vec{a} = (2, 5, 2)$
$\vec{b} = (3, 3, -4)$

Solution:

$|\vec{a}| = 5.7446$

$|\vec{b}| = 5.8310$

$\vec{a} \cdot \vec{b} = 13$

$\theta = 67.1635°$

Keystrokes	Display
[f] [REG] [f] [PRGM]	
2 [ENTER↑] 3 [R/S]	
5 [ENTER↑] 3 [R/S]	
2 [ENTER↑] 4 [CHS] [R/S]	
[RCL] 0 [f] [√x]	5.7446
[RCL] 1 [f] [√x]	5.8310
[RCL] 2	13.0000
[GSB] 13	67.1635

Numerical Methods

Newton's Method Solution to f(x) = 0

This program uses Newton's method to find a solution for f(x) = 0, where f(x) is specified by the user.

The user must define the function f(x) by keying into program memory the keystrokes required to find f(x), assuming x is in the X-register. 20 program steps are available for defining f(x); the program only uses registers R_0 through R_4, the rest of the registers are available to the user.

The user must provide the program with an initial guess, x_1, for the solution. The closer the initial guess is to the actual solution, the faster the program will converge to an answer. The program will halt when two successive approximations for x, say x_i and x_{i+1}, are within a tolerance ϵ, i.e., when $|x_{i+1} - x_i| < \epsilon$. The value for ϵ must be input by the user. In general a reasonable value for ϵ might be $10^{-6}\ x_1$.

Equations:

The basic formula used by Newton's method to generate the next approximation for the solution is

$$x_{i+1} = x_i - \frac{f\ (x_i)}{f'\ (x_i)}$$



This program makes a numerical approximation for the derivative f′ (x) to give the following equation:

$$x_{i+1} = x_i - \delta_i \left[\frac{f(x_i + \delta_i)}{f(x_i)} - 1 \right]^{-1}$$

where: $\delta_i = 10^{-5} x_i$

Notes:

1. After the routine has finished calculating, the last value of f(x) may be displayed by pressing RCL 4. If this value is not close enough to zero, the program may be run again with a smaller value for ϵ.

2. The user can watch the function converge to zero by making a slight change in the program. If the g NOP in line 45 is replaced by an f PAUSE, the program will pause during each iteration, displaying successive values of f(x) which should be converging to zero. To make this change to a program that has already been keyed in, perform the following operations:

    1. Press GTO 44
    2. Switch to PRGM
    3. Press f PAUSE
    4. Switch to RUN
    5. Press f PRGM

KEY ENTRY	DISPLAY
f CLEAR PRGM	00
RCL 1	01– 24 1
GSB 25	02– 12 25
STO 4	03– 23 4
RCL 1	04– 24 1
RCL 1	05– 24 1
EEX	06– 33
5	07– 5
÷	08– 71
STO 3	09– 23 3
+	10– 51
GSB 25	11– 12 25
RCL 4	12– 24 4
÷	13– 71
1	14– 1
−	15– 41

KEY ENTRY	DISPLAY
g 1/x	16– 15 3
RCL 3	17– 24 3
×	18– 61
STO − 1	19– 23 41 1
g ABS	20– 15 34
RCL 2	21– 24 2
f x≤y	22– 14 41
GTO 01	23– 13 01
GTO 49	24– 13 49
g NOP	45– 15 13
g x=0	46– 15 71
GTO 49	47– 13 49
g RTN	48– 15 12
RCL 1	49– 24 1

REGISTERS			
R_0	R_1 x	R_2 ϵ	R_3 δ
R_4 f(x)	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in lines 1-24 of the			
	program			24- 13 49
2	Key in function f(x)			
3	Key in a branch to line 45		[GTO] 45	
4	Press [SST] until display			
	shows line 44			
5	Key in lines 45-49 of program			
6	Switch to RUN			
7	Store initial guess for solution	x_1	[STO] 1	
8	Store tolerance	ϵ	[STO] 2	
9	Calculate solution		[GSB] 01	x_0
10	To change x_1 or ϵ go to			
	appropriate step and store			
	new value.			

Example:

An equation often solved by gear designers is

$$\tan x - x - I = 0$$

where x is an angle in radians and I is the *involute* of x. Find the angle x_0 corresponding to an involute of 0.0324.

Note:

Since a gear designer might want to calculate x for several values of I, it will be simpler to store I in R_7 for use by the function f(x).

Solution:

$x_0 = 25.62°$
Last $f(x) = 1.30 \times 10^{-9}$

Keystrokes	**Display**
Switch to PRGM	
(Key in lines 1-24 of the program)	
f TAN f LST x	
− RCL 7 −	
GTO 45 SST	
... (until display shows line 44)	
(Key in lines 45-49 of the program)	
Switch to RUN	
g RAD .0324	
STO 7	
1 STO 1	
EEX CHS 6 STO 2	
GSB 01	0.4472
180 × g π ÷	25.6211
RCL 4	1.3000−09

Numerical Integration, Simpson's Rule

Let x_0, x_1, ... x_n be equally spaced points such that $x_i = x_0 + ih$ for $i = 0, 1, 2, \ldots, n$ at which corresponding values $f(x_0)$, $f(x_1)$, ... $f(x_n)$ of a function f(x) are known. This function need not be known explicitly but if it is, these values can be found previously by writing the function into memory and evaluating at the various points. n must be an even positive integer.

$$\int_{x_0}^{x_n} f(x)\,dx \simeq \frac{h}{3}\left[f(x_0) + 4f(x_1) + 2f(x_2) + \ldots + 4f(x_{n-3}) + 2f(x_{n-2}) + 4f(x_{n-1}) + f(x_n)\right].$$

Let the solution be indicated by I.

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	GSB 18	12– 12 18
RCL 0	01– 24 0	R/S	13– 74
3	02– 3	2	14– 2
÷	03– 71	×	15– 61
STO 0	04– 23 0	GSB 18	16– 12 18
×	05– 61	GTO 09	17– 13 09
STO 1	06– 23 1	RCL 0	18– 24 0
R/S	07– 74	×	19– 61
GSB 18	08– 12 18	STO + 1	20– 23 51 1
R/S	09– 74	RCL 1	21– 24 1
4	10– 4	g RTN	22– 15 12
×	11– 61		

REGISTERS			
R_0 h/3	R_1 Σ	R_2	R_3
R_4	R_5	R_6	R_7

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Store increment	h	STO 0	
3	Enter first function value	$f(x_0)$	GSB 01	Partial sum
4	Enter last function value	$f(x_n)$	R/S	Partial sum
5	Enter values i = 1, 2,..., n −2	$f(x_i)$	R/S	Partial sum
6	Enter value i = n – 1	$f(x_{n-1})$	R/S	I

Example:

Compute $\int_0^{\pi} \sin^2 x \, dx$ using Simpson's rule with $h = \pi/8$.

The following data must be found first:

i	0	1	2	3	4	5	6	7	8
x_i	0	$\pi/8$	$\pi/4$	$3\pi/8$	$\pi/2$	$5\pi/8$	$3\pi/4$	$7\pi/8$	π
$f(x_i)$	0	0.1464	0.5	0.8536	1	0.8536	0.5	0.1464	0

Solution:

$$\int_0^{\pi} \sin^2 x \, dx \simeq 1.5708$$

The exact solution is $\pi/2$.

Keystrokes	**Display**
[g] [π] 8 [÷] [STO] 0	
0 [GSB] 01	0.0000
0 [R/S]	0.0000
0.1464 [R/S] 0.5	
[R/S] 0.8536 [R/S] 1	
[R/S] 0.8536 [R/S]	
0.5 [R/S] 0.1464 [R/S]	1.5708

Analytical Geometry

Hyperbolic Functions

This program evaluates the six hyperbolic functions by the following formulas:

1. $\sinh x = \dfrac{e^x - e^{-x}}{2}$

2. $\cosh x = \dfrac{e^x + e^{-x}}{2}$

3. $\tanh x = \dfrac{e^x - e^{-x}}{e^x + e^{-x}}$

4. $\operatorname{csch} x = \dfrac{1}{\sinh x}$ $(x \neq 0)$

5. $\operatorname{sech} x = \dfrac{1}{\cosh x}$

6. $\coth x = \dfrac{1}{\tanh x}$ $(x \neq 0)$

KEY ENTRY	DISPLAY
f CLEAR PRGM	00
g e^x	01– 15 1
ENTER↑	02– 31
g 1/x	03– 15 3
−	04– 41
2	05– 2
÷	06– 71
GTO 00	07– 13 00
g e^x	08– 15 1
ENTER↑	09– 31
g 1/x	10– 15 3
+	11– 51
GTO 05	12– 13 05

KEY ENTRY	DISPLAY
g e^x	13– 15 1
ENTER↑	14– 31
g 1/x	15– 15 3
−	16– 41
ENTER↑	17– 31
ENTER↑	18– 31
f LST x	19– 14 73
2	20– 2
×	21– 61
+	22– 51
÷	23– 71
GTO 00	24– 13 00

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	sinh x	x	[GSB] 01	sinh x
	or			
	cosh x	x	[GSB] 08	cosh x
	or			
	tanh x	x	[GSB] 13	tanh x
	or			
	csch x	x	[GSB] 01	
			[g] [1/x]	csch x
	or			
	sech x	x	[GSB] 08	
			[g] [1/x]	sech x
	or			
	coth x	x	[GSB] 13	
			[g] [1/x]	coth x

Examples:

1. sinh 2.5 = 6.0502
2. cosh 3.2 = 12.2866
3. tanh 1.9 = 0.9562
4. csch 4.6 = 0.0201
5. sech (−.25) = 0.9695
6. coth (−2.01) = −1.0366

Keystrokes	Display
2.5 GSB 01	6.0502
3.2 GSB 08	12.2866
1.9 GSB 13	0.9562
4.6 GSB 01 g 1/x	0.0201
.25 CHS GSB 08	
g 1/x	0.9695
2.01 CHS GSB 13	
g 1/x	-1.0366

Inverse Hyperbolic Functions

This program evaluates the inverse hyperbolic functions by the following formulas:

1. $\sinh^{-1} x = \ln\left[x + (x^2 + 1)^{1/2}\right]$

2. $\cosh^{-1} x = \ln\left[x + (x^2 - 1)^{1/2}\right]$ $\quad x \geqslant 1$

3. $\tanh^{-1} x = \frac{1}{2} \ln\left[\frac{1 + x}{1 - x}\right]$ $\quad x^2 < 1$

4. $\operatorname{csch}^{-1} x = \sinh^{-1}\left[\frac{1}{x}\right]$ $\quad x \neq 0$

5. $\operatorname{sech}^{-1} x = \cosh^{-1}\left[\frac{1}{x}\right]$ $\quad 0 < x \leqslant 1$

6. $\coth^{-1} x = \tanh^{-1}\left[\frac{1}{x}\right]$ $\quad x^2 > 1$

KEY ENTRY	DISPLAY	
f CLEAR PRGM	00	
ENTER↑	01–	31
ENTER↑	02–	31
×	03–	61
1	04–	1
+	05–	51
f √x	06–	14 0
+	07–	51
f LN	08–	14 1
GTO 00	09–	13 00
ENTER↑	10–	31
ENTER↑	11–	31
×	12–	61
1	13–	1
−	14–	41
f √x	15–	14 0

KEY ENTRY	DISPLAY	
+	16–	51
f LN	17–	14 1
GTO 00	18–	13 00
ENTER↑	19–	31
ENTER↑	20–	31
1	21–	1
+	22–	51
x⇄y	23–	21
CHS	24–	32
1	25–	1
+	26–	51
÷	27–	71
f LN	28–	14 1
2	29–	2
÷	30–	71
GTO 00	31–	13 00

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	$\sinh^{-1} x$	x	[GSB] 01	$\sinh^{-1} x$
	or			
	$\cosh^{-1} x$	x	[GSB] 10	$\cosh^{-1} x$
	or			
	$\tanh^{-1} x$	x	[GSB] 19	$\tanh^{-1} x$
	or			
	$\text{csch}^{-1} x$	x	[g] [1/x] [GSB] 01	$\text{csch}^{-1} x$
	or			
	$\text{sech}^{-1} x$	x	[g] [1/x] [GSB] 10	$\text{sech}^{-1} x$
	or			
	$\coth^{-1} x$	x	[g] [1/x] [GSB] 19	$\coth^{-1} x$

Example:

1. $\sinh^{-1} (2.4) = 1.6094$
2. $\cosh^{-1} (90) = 5.1929$
3. $\tanh^{-1} (-.65) = -0.7753$
4. $\text{csch}^{-1} (2) = 0.4812$
5. $\text{sech}^{-1} (.4) = 1.5668$
6. $\coth^{-1} (3.4) = 0.3031$

Keystrokes	Display
2.4 [GSB] 01	*1.6094*
90 [GSB] 10	*5.1929*
.65 [CHS] [GSB] 19	*-0.7753*
2 [g] [1/x] [GSB] 01	*0.4812*
.4 [g] [1/x] [GSB] 10	*1.5668*
3.4 [g] [1/x] [GSB] 19	*0.3031*

Circle Determined by Three Points

This program calculates the center (x_0, y_0) and radius (r) of a circle given three non-collinear points.

The equation of a circle is:

$$(x - x_0)^2 + (y - y_0)^2 = r^2$$

x_0 and y_0 are solved from:

$$\begin{bmatrix} (x_1 - x_3) & (y_1 - y_3) \\ (x_1 - x_2) & (y_1 - y_2) \end{bmatrix} \begin{bmatrix} x_0 \\ y_0 \end{bmatrix} =$$

$$\begin{bmatrix} \tfrac{1}{2}\left[(x_1 - x_3)(x_1 + x_3) + (y_1 - y_3)(y_1 + y_3)\right] \\ \tfrac{1}{2}\left[(x_1 - x_2)(x_1 + x_2) + (y_1 - y_2)(y_1 + y_2)\right] \end{bmatrix}$$

and $r = \sqrt{(x_1 - x_0)^2 + (y_1 - y_0)^2}$

KEY ENTRY	DISPLAY	KEY ENTRY	DISPLAY
f CLEAR PRGM	00	RCL 4	24– 24 4
ENTER↑	01– 31	RCL 5	25– 24 5
RCL 1	02– 24 1	×	26– 61
−	03– 41	−	27– 41
CHS	04– 32	STO 1	28– 23 1
R/S	05– 74	RCL 7	29– 24 7
x⇄y	06– 21	RCL 3	30– 24 3
RCL 1	07– 24 1	×	31– 61
+	08– 51	RCL 6	32– 24 6
×	09– 61	RCL 4	33– 24 4
R/S	10– 74	GSB 41	34– 12 41
CLX	11– 34	R/S	35– 74
RCL 0	12– 24 0	RCL 6	36– 24 6
−	13– 41	RCL 2	37– 24 2
CHS	14– 32	×	38– 61
R/S	15– 74	RCL 5	39– 24 5
x⇄y	16– 21	RCL 3	40– 24 3
RCL 0	17– 24 0	×	41– 61
+	18– 51	−	42– 41
×	19– 61	2	43– 2
R/S	20– 74	÷	44– 71
RCL 7	21– 24 7	RCL 1	45– 24 1
RCL 2	22– 24 2	÷	46– 71
×	23– 61	g RTN	47– 15 12

REGISTERS			
R_0 x_1	R_1 y_1, Det	R_2 $(y_1 - y_2)$	R_3 used
R_4 $(x_1 - x_2)$	R_5 $(y_1 - y_3)$	R_6 used	R_7 $(x_1 - x_3)$

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Store (x_1, y_1)	x_1	[STO] 0	
		y_1	[STO] 1	
3	Input (x_2, y_2) and calculate	x_2	[ENTER↑]	
		y_2	[ENTER↑]	
			[GSB] 01	
			[STO] 2 [R/S]	
			[STO] 3 [R/S]	
			[STO] 4 [R/S]	
			[STO] [+] 3	
4	Input (x_3, y_3) and calculate	x_3	[ENTER↑]	
		y_3	[ENTER↑]	
			[GSB] 01	
			[STO] 5 [R/S]	
			[STO] 6 [R/S]	
			[STO] 7 [R/S]	
			[STO] [+] 6	
5	Calculate y_0		[GSB] 21	y_0
6	Calculate x_0		[R/S]	x_0
7	Calculate r	x_1	[−] [x⇄y]	
		y_1	[−] [g] [→P]	r
	NOTE: The stack should be			
	maintained between step 2			
	to step 7.			

Example 1:

Find the equation of the circle that goes through the three points (1, 1) (3.5, −7.6), (12, 0.8).

Solution 1:

Center = (6.45, −2.08), r = 6.26

Equation: $(x - 6.45)^2 + (y + 2.08)^2 = (6.26)^2$

Keystrokes	Display
[f] [FIX] 2	
1 [STO] 0	
1 [STO] 1	
3.5 [ENTER↑] 7.6	
[CHS] [ENTER↑]	
[GSB] 01	8.60
[STO] 2	
[R/S] [STO] 3	
[R/S] [STO] 4	
[R/S] [STO] [+] 3	−11.25
12 [ENTER↑] .8	
[ENTER↑] [GSB] 01	0.20
[STO] 5	
[R/S] [STO] 6	
[R/S] [STO] 7	
[R/S] [STO] [+] 6	−143.00
[GSB] 21	−2.08
[R/S]	6.45
1 [−] [x⇄y] 1	
[−] [g] [→P]	6.26

Example 2:

Find the equation of the circle that passes through the three point (0, 1), (−1, 0), (0, −1).

Solution 2:

Center = (0, 0), r = 1

Equation: $x^2 + y^2 = 1$

Keystrokes	Display
[f] [FIX] 2	
0 [STO] 0	
1 [STO] 1	
1 [CHS] [ENTER↑] 0	
[ENTER↑] [GSB] 01	1.00
[STO] 2	
[R/S] [STO] 3	
[R/S] [STO] 4	
[R/S] [STO] [+] 3	-1.00
0 [ENTER↑] 1 [CHS]	
[ENTER↑] [GSB] 01	2.00
[STO] 5	
[R/S] [STO] 6	
[R/S] [STO] 7	
[R/S] [STO] [+] 6	0.00
[GSB] 21	0.00
[R/S]	0.00
0 [−] [x⇄y] 1	
[−] [g] [→P]	1.00

Intersection of Line and Line

This program calculates the point of intersection (x_p, y_p) of two lines, each of them is specified by two points, i.e., (x_1, y_1), (x_1', y_1') and (x_2, y_2), (x_2', y_2'). x_p and y_p are calculated from the equation of straight line $y = x \tan \theta + c$. i.e., by solving:

$$\begin{bmatrix} 1 & -\tan \theta_1 \\ 1 & -\tan \theta_2 \end{bmatrix} \begin{bmatrix} y_p \\ x_p \end{bmatrix} = \begin{bmatrix} c_1 \\ c_2 \end{bmatrix}$$

"Error 0" will be displayed if two lines have the same slope, i.e., if there are either parallel or overlapping.

KEY ENTRY	DISPLAY
f CLEAR PRGM	00
GSB 37	01– 12 37
STO 3	02– 23 3
RCL 2	03– 24 2
RCL 1	04– 24 1
RCL 3	05– 24 3
×	06– 61
−	07– 41
STO 4	08– 23 4
R/S	09– 74
GSB 37	10– 12 37
STO 5	11– 23 5
RCL 2	12– 24 2
RCL 1	13– 24 1
RCL 5	14– 24 5
×	15– 61
−	16– 41
STO 6	17– 23 6
RCL 4	18– 24 4
−	19– 41
RCL 3	20– 24 3
RCL 5	21– 24 5
−	22– 41
STO 7	23– 23 7
÷	24– 71
R/S	25– 74
RCL 4	26– 24 4
RCL 5	27– 24 5
CHS	28– 32
×	29– 61
RCL 3	30– 24 3
RCL 6	31– 24 6
×	32– 61
+	33– 51
RCL 7	34– 24 7
÷	35– 71
R/S	36– 74
STO 1	37– 23 1
−	38– 41
STO 0	39– 23 0
R↓	40– 22
STO 2	41– 23 2
−	42– 41
RCL 0	43– 24 0
g →P	44– 15 4
R↓	45– 22
f TAN	46– 14 9
g RTN	47– 15 12

REGISTERS			
R_0 used	R_1 x_1', x_2'	R_2 y_1', y_2'	R_3 tan θ_1
R_4 c_1	R_5 tan θ_2	R_6 c_2	R_7 Det

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Key in the program			
2	Input the first line by two			
	points	x_1	ENTER↑	
		x_1'	ENTER↑	
		y_1	ENTER↑	
		y_1'	GSB 01	c_1
3	Input the 2nd line by two			
	points and calculate the			
	intersection	x_2	ENTER↑	
		x_2'	ENTER↑	
		y_2	ENTER↑	
		y_2'	GSB 10	x_p
			R/S	y_p
4	For a new pair of lines, go to			
	step 2.			

Example:

Find the intersection of the lines defined by the points (4, 8), (−1, −2) and (−3, 9), (7, −1).

Solution:

$x_p = 4,\ y_p = 2$

Keystrokes	Display
4 ENTER↑ 1	
CHS ENTER↑ 8	
ENTER↑ 2 CHS	
GSB 01	0.0000
3 CHS ENTER↑ 7	
ENTER↑ 9 ENTER↑ 1	
CHS GSB 10	4.0000
R/S	2.0000

NOTES

HEWLETT PACKARD

1000 N.E. Circle Blvd., Corvallis, OR 97330

For additional sales and service information contact your local Hewlett-Packard Sales Office or call 800/648-4711. (In Nevada call 800/992-5710.)

00033-90030

Printed in U.S.A.